Problem 7.1 − Random Systems: Cumulants

Generate right− and left−hand−side of equation 7.25.

```
numTerms = 10;
```

$$\texttt{LHS} = \texttt{Expand}\left[\texttt{Normal}[\texttt{Series}[e^x, \{x, 0, \texttt{numTerms}\}]] \,/.\, x \to \sum_{n=1}^{\texttt{numTerms}} \frac{(i\,k)^n\, c[n]}{n!}\right];$$

$$\texttt{RHS} = \sum_{n=1}^{\texttt{numTerms}} \frac{(i\,k)^n\, m[n]}{n!};$$

Solve for moments in terms of cumulants.

```
moment[n_] :=
  (moment[n] = (m[n] /. Solve[Coefficient[LHS, k^n] == Coefficient[RHS, k^n], m[n]])[[1]]);
```

```
moment /@ Range[3]
```

$\{c[1],\ c[1]^2 + c[2],\ c[1]^3 + 3\,c[1]\,c[2] + c[3]\}$

Solve for cumulants in terms of moments.

```
cumulant[n_] := (cumulant[n] =
    (c[n] /. Solve[m[n] == (moment[n] /. c[k_?(# < n &)] :> cumulant[k]), c[n]])[[1]]);
```

First four cumulants:

```
TableForm[c[#] == cumulant[#] & /@ Range[4]]
```

$c[1] == m[1]$

$c[2] == -m[1]^2 + m[2]$

$c[3] == 2\,m[1]^3 - 3\,m[1]\,m[2] + m[3]$

$c[4] == -6\,m[1]^4 + 12\,m[1]^2\,m[2] - 3\,m[2]^2 - 4\,m[1]\,m[3] + m[4]$

First four cumulants for a Gaussian:

$$p[x\_] := \frac{e^{-\frac{(x-\mu)^2}{2\sigma^2}}}{\sqrt{2\pi\sigma^2}};$$

$$\texttt{Assuming}\Big[\{\mu > 0,\ \sigma > 0\},\ \texttt{TableForm}\Big[\texttt{Simplify}\Big[\{c[p, \#], \texttt{"="}, \texttt{cumulant}[\#]\} \,/.\, m[n\_] :> \int_{-\infty}^{\infty} x^n\, p[x]\, dx\Big] \,\&\, /@\, \texttt{Range}[4]\Big]\Big]$$

c[p, 1]	=	μ
c[p, 2]	=	σ^2
c[p, 3]	=	0
c[p, 4]	=	0

Problem 7.2 − Random Systems: 2D Transform

The equations given.

```
y1 = Sqrt[-2 Log[x1]] Sin[x2];
y2 = Sqrt[-2 Log[x1]] Cos[x2];
```

$$y1 = \sqrt{-2\,\text{Log}[x1]}\ \text{Sin}[x2];\qquad y2 = \sqrt{-2\,\text{Log}[x1]}\ \text{Cos}[x2];$$

The matrix determinant of the derivatives produces the area of the parallelogram. The reciprocal is p(y1,y2).

$$\left(1\Big/\text{Det}\left[\begin{pmatrix} \text{D}[y1,\ x1] & \text{D}[y2,\ x1] \\ \text{D}[y1,\ x2] & \text{D}[y2,\ x2] \end{pmatrix}\right]\right)\ //\ \text{Simplify}$$

```
x1
```

We find an expression for p(y1,y2). (This needs to be multiplied by the normalized uniform probably of (x1,x2).)

```
x1 /. Solve[{Y1 == y1, Y2 == y2}, {x1, x2}][[1]]
```

```
Solve::ifun : Inverse functions are being used by Solve, so some
   solutions may not be found; use Reduce for complete solution information. More…
```

$$e^{\frac{1}{2}\left(-Y1^2 - Y2^2\right)}$$

Here's a histogram of (y1,y2) with (x1,x2) drawn from a uniform distribution, $0<x1<1$, $0<x2<2\pi$.

```
random[] :=
  Function[{x1, x2}, Evaluate[{y1, y2}]][Random[Real, {0, 1}], Random[Real, {0, 2 π}]];

Needs["Graphics`Graphics`"];
Needs["Graphics`Graphics3D`"];
Histogram3D[Table[random[], {5000}]];
```



Problem 7.3 − Random Systems: LFSR

Here is a program to generate the next bit of an LFSR.

```
nextBit[taps_][bits_] := Prepend[bits, Mod[Total[bits〚taps〛], 2]];
```

Here are 60 bits of the LFSR output.

```
output = Nest[nextBit[{1, 4}], {0, 0, 0, 1}, 56]
```

```
{0, 0, 1, 1, 0, 1, 0, 1, 1, 1, 1, 0, 0, 0, 1, 0, 0, 1, 1, 0, 1, 0, 1, 1, 1, 1, 0, 0, 0, 1, 0,
 0, 1, 1, 0, 1, 0, 1, 1, 1, 1, 0, 0, 0, 1, 0, 0, 1, 1, 0, 1, 0, 1, 1, 1, 1, 0, 0, 0, 1}
```

These 60 bits taken 4 at a time make each number from 1 to 15 exactly once.

```
Sort[FromDigits[#, 2] & /@ Partition[output, 4]]
```

```
{1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15}
```

Here are the number of unrepeated bits for a given register size.

```
unrepeatedBits[registerSize_] := (2^registerSize - 1) * registerSize;
```

Here is the time in years it takes to run through these numbers on a 1 GHz machine.

```
Needs["Miscellaneous‘Units‘"]
norepeatTime[registerSize_] :=
  Convert[unrepeatedBits[registerSize] / (1 Giga Hertz), Year] / Year;
```

Below is a semilog plot showing that a register size of about 80 is needed to last the age of the universe (10^10 years).

```
Needs["Graphics‘Graphics‘"];
LogPlot[{10^10, norepeatTime[n]}, {n, 10, 100}];
```